

В системе «1С:Предприятие 8» имеется встроенный механизм регистрации пользователей информационной базы. Ведение этого списка пользователей осуществляется в конфигураторе. Если в информационной базе не заведено ни одного пользователя, как сейчас у нас, то вход в программу осуществляется в режиме неавторизованного пользователя.

Нам необходимо сейчас разработать такой механизм, чтобы при запуске программы от имени определенного пользователя информационной базы (ИБ) в справочнике *Пользователи* был найден соответствующий данному пользователю ИБ элемент. Если такого элемента нет, то он должен быть создан.

При добавлении нового пользователя мы должны будем, помимо всего прочего, указать доступные ему роли, которые описывают полномочия различных категорий пользователей на доступ к информации. Новую роль можно создать в ветви дерева конфигурации *Общие – Роли*.

Каждая роль описывает набор прав на выполнение каких-либо действий с информацией базы данных.


Роли могут быть заданы в достаточно широких пределах — от возможности только просмотра ограниченного числа видов объектов до полного набора прав по вводу, просмотру, корректировке и удалению любых видов данных.

В системе обязательно должна быть роль с административными правами.


## Задание 1

### Разработайте справочник Пользователи.


1. Создадим справочник – *Пользователи* :




2. На закладке подсистемы укажем его принадлежность подсистеме *Состав школы*:




3. На закладке *Данные* укажем длину наименования равной 150:



4. Теперь нажатием на пиктограмму  запустим программу, как это показано на рисунке:




В подсистеме *Состав школы* откроем справочник *Пользователи*. Вначале этот справочник пустой. Добавим в него двух учеников, как это показано на рисунке:



## Задание 2


**Разработайте в информационной базе *Наша школа* роли различных пользователей.**

1. Запустим программу в режиме *Конфигуратор*. Раскроем ветвь дерева конфигурации *Общие* и добавим новую *Роль* :




2. Зададим имя этой роли – *ПолныеПрава*, синоним – *Полные права*.

В открывшемся окне редактирования прав установим для этой роли все права на все объекты конфигурации и поставим флаг *Устанавливать права для новых объектов* :



3. Аналогично добавим еще две роли: *Учитель* и *Ученик* :




5. Обновим конфигурацию (F7) или запустим информационную базу в режиме отладки (F5).

На последующих занятиях мы посмотрим, как можно с помощью механизма настройки прав регулировать доступ пользователей к различным объектам информационной базы. Пока оставим у всех полные права. Сохраним конфигурацию.


### Задание 3

**Разработайте список пользователей.**


1. Откроем список пользователей, для этого выберем в главном меню пункт *Администрирование – Пользователи* :



2. В появившемся окне добавим пользователя, которого назовем *Администратор* :



3. На закладке *Прочие* укажем для этого пользователя роль *Полные права*, язык – *Русский* :



4. Добавим еще одного пользователя: *Имя* – Смирнова, *Полное имя* – Смирнова Евгения Владимировна, на закладке *Прочие* укажем роль – *Ученик* :

Пользователь

Основные Прочие

Имя: Смирнова

Полное имя: Смирнова Евгения Владимировна

☒ Аутентификация 1С:Предприятия:

Пароль:

Подтверждение пароля:

Пользователю запрещено изменять пароль ☐


Показывать в списке выбора ☒

☐ Аутентификация операционной системы:


Пользователь:

ОК Отмена Справка

5. Теперь немного изменим созданный ранее справочник *Пользователи*. Найдём этот справочник в дереве конфигурации, развернём его свойства и добавим реквизит:



6. В открывшемся окне свойств укажем имя реквизита – *ИдентификаторПользователяИБ*, зададим тип – *УникальныйИдентификатор*, укажем свойство *Индексировать*, как это показано на рисунке:



По этому реквизиту мы будем определять соответствие элемента справочника *Пользователи* пользователю информационной базы.


#### Задание 4

**Разработайте процедуры для определения текущего пользователя.**

1. Значение элемента справочника *Пользователи* нам необходимо будет сохранять во время всего сеанса работы пользователя, для таких случаев в конфигурации предусмотрен тип объекта, называемый *Параметры сеанса*. В дереве конфигурации объекты данного типа расположены в ветви *Общие*. Также как и глобальные переменные, параметры сеанса доступны в любом программном модуле конфигурации. Основные отличия параметров сеанса:


- параметры сеанса являются объектами метаданных, что позволяет «1С:Предприятию» осуществлять повышенный контроль за их использованием;
- параметры сеанса имеют определенный тип;
- для установки или получения значения параметра сеанса текущий пользователь должен быть наделен соответствующим правом.

Добавим параметр сеанса *ТекущийПользователь*, в свойствах укажем тип *СправочникСсылка.Пользователи* :




2. Напишем процедуру, которая будет искать в справочнике Пользователи необходимый элемент, а в случае его отсутствия – создавать. Найденный или созданный элемент будет сохранять в параметре *ТекущийПользователь*.

Поместим эту процедуру в общий модуль, который назовем *Пользователи*. Раскроем в дереве конфигурации ветвь *Общие*, найдем *Общие модули* и добавим новый модуль:



3. В свойствах модуля укажем *Вызов сервера* и *Привилегированный*.





#### 4. В тексте модуля поместим процедуру:

```
// Процедура, определяет пользователя, под которым запущен сеанс и пытается
// найти соответствие ему в справочнике Пользователи. Если соответствие
// не найдено - создается новый элемент. Параметр сеанса ТекущийПользователь
// устанавливается как ссылка на найденный (созданный) элемент справочника.
//
Процедура ОпределитьТекущегоПользователя() Экспорт

    Если ПользователиИнформационнойБазы.ПолучитьПользователей().Количество() = 0 Тогда
        // Не добавлено ни одного пользователя ИБ. Запрещаем работу с программой!
        ТекстСообщенияОбОшибке = "Пользователь не определен!";
        ВызватьИсключение ТекстСообщенияОбОшибке;
    Иначе
        ТекущийПользовательИБ = ПользователиИнформационнойБазы.ТекущийПользователь();
        ИдентификаторПользователяИБ = ТекущийПользовательИБ.УникальныйИдентификатор;

        ТекущийПользователь = Справочники.Пользователи.НайтиПоРеквизиту(
            "ИдентификаторПользователяИБ",
            ИдентификаторПользователяИБ);

        Если ТекущийПользователь.Пустая() Тогда

            НовыйПользователь = Справочники.Пользователи.СоздатьЭлемент();
            НовыйПользователь.ИдентификаторПользователяИБ = ИдентификаторПользователяИБ;
            НовыйПользователь.Наименование = ТекущийПользовательИБ.ПолноеИмя;

            НовыйПользователь.Записать();

            ТекущийПользователь = НовыйПользователь.Ссылка;

        КонецЕсли;

        ПараметрыСеанса.ТекущийПользователь = ТекущийПользователь;

    КонецЕсли;

КонецПроцедуры
```

#### 5. В этот же модуль поместим функцию:


```
Функция ТекущийПользователь() Экспорт

    Возврат ПараметрыСеанса.ТекущийПользователь;

КонецФункции
```

6. Для инициализации параметров сеанса предназначен *Модуль сеанса*, который автоматически выполняется при старте системы «1С:Предприятие», в момент загрузки конфигурации. Установка параметров сеанса выполняется в обработчике события: *УстановкаПараметровСеанса()*.

Откроем этот модуль:



#### 7. Поместим в него процедуру:


```
Процедура УстановкаПараметровСеанса(ТребуемыеПараметры)

    Пользователи.ОпределитьТекущегоПользователя();

КонецПроцедуры
```

Теперь при запуске системы автоматически запустится процедура *УстановкаПараметровСеанса*, которая в свою очередь запустит процедуру общего модуля. Процедура *ОпределитьТекущегоПользователя* попытается найти или создать в справочнике Пользователи подходящий элемент и запишет значение ссылки на этот элемент в параметр сеанса *ТекущийПользователь*.

8. Для проверки данного механизма откроем модуль управляемого приложения:



Данный модуль выполняется автоматически в момент загрузки конфигурации после модуля сеанса. Модуль управляемого приложения предназначен для отработки действий, связанных с сеансом работы пользователя (прежде всего обработки начала и окончания сеанса работы). В нем рекомендуется реализовывать только обработчики соответствующих событий.

9. Поместим в модуль управляемого приложения две процедуры, которые являются обработчиками событий:

- *ПередНачаломРаботыСистемы* – выполняется при запуске программы в режиме «1С:Предприятие» до открытия главного окна. В процедуре-обработчике данного события могут быть выполнены необходимые проверки и при необходимости установлен параметр отказа от запуска программы;
- *ПриНачалеРаботыСистемы* – выполняется при запуске в режиме «1С:Предприятие» после открытия главного окна. В процедуре-обработчике могут быть выполнены различные действия, необходимые при запуске программы, например, открыты некоторые формы.

*Перем мТекущийПользователь Экспорт;*

```
Процедура ПередНачаломРаботыСистемы(Отказ)
    мТекущийПользователь = Пользователи.ТекущийПользователь();
КонецПроцедуры


Процедура ПриНачалеРаботыСистемы(Отказ)
    УстановитьЗаголовокПриложения( "Наша школа (Пользователь: "
        + мТекущийПользователь + ")");
КонецПроцедуры
```

В этом модуле мы объявили глобальную переменную *мТекущийПользователь*, затем в обработчике события *ПередНачаломРаботыСистемы* присвоили этой переменной значение текущего пользователя. Мы вынуждены были обратиться к процедуре общего модуля *Пользователи*, поскольку в модуле управляемого приложения недоступно свойство глобального контекста *ПараметрыСеанса*.


Для чего необходимо хранить одну и ту же информацию в глобальной переменной и в параметрах сеанса? Дело в том, что параметры сеанса и глобальная переменная, объявленная в модуле управляемого приложения, доступны в различных процедурах и функциях конфигурации. В одних


случаях мы будем определять текущего пользователя через параметры сеанса, в других – через переменную управляемого приложения. Параметры сеанса будут доступны в процедурах и функциях, которые выполняются на сервере, переменная – в клиентской части программы. Подробнее о том, что такое серверная и клиентская часть программы, мы рассмотрим в следующих практических занятиях.

### Схема взаимодействия модулей



10. В обработке события *ПриНачалеРаботыСистемы* мы заменили стандартный заголовок системы и теперь сможем видеть, под каким пользователем запущена программа:





Запустив программу в режиме «1С:Предприятие», поочередно под разными пользователями, мы можем убедиться, что эти пользователи были успешно добавлены в справочник.

### Скачать листинг

Если Ваша информационная база, по каким-то причинам некорректно работает, то сохраните файл с **выгрузкой** и загрузите информационную базу: в режиме конфигуратора откройте в главном меню *Администрирование – Загрузить информационную базу* – укажите путь к сохраненному файлу.